

---

Subject: Re: widgets

Posted by [David Fanning](#) on Fri, 19 Aug 2005 15:02:59 GMT

[View Forum Message](#) <> [Reply to Message](#)

---

hernan writes:

> Como puedo hacer un widget que sume dos valores?

Como esto:

```
pro widget_sum_event, event
```

```
widget_control, event.top, get_uvalue = info
widget_control, info.text1, get_value=value1
widget_control, info.text2, get_value=value2
value = Float(value1[0]) + Float(value2[0])
widget_control, info.text, set_value = StrTrim(string(value),2)
```

```
end
```

```
pro widget_sum
```

```
;* Ingresamos la base de trabajo.
;* Creamos la primera entrada de datos
base_suma = widget_base(xsize = 300, ysize = 200)
eti1 = widget_label(base_suma, value = 'Ingrese el valor de a: ', $
xsize = 100, xoffset = 20, yoffset = 20)
text1 = widget_text(base_suma, value = '1', uvalue = 'uno', $
xoffset = 120, yoffset = 20, /editable)
;* Creamos la segunda entrada de datos
eti2 = widget_label(base_suma, value = 'Ingrese el valor de b: ', $
xsize = 100, xoffset = 20, yoffset = 60)
text2 = widget_text(base_suma, value = '1', uvalue = 'dos', $
xoffset = 120, yoffset = 60, /editable)
;* Creamos la salida de los datos
eti = widget_label(base_suma, value = 'La suma es: ', $
xsize = 100, xoffset = 20, yoffset = 100)
text = widget_text(base_suma, value = '1', $
xoffset = 120, yoffset = 100, /editable)
widget_control, base_suma, set_uvalue = 'text'
info = {text1:text1, text2:text2, text:text}
widget_control, base_suma, /realize, set_uvalue=info
xmanager, 'widget_sum', base_suma, /no_block
end
```

Salut,

David

--

David Fanning, Ph.D.

Fanning Software Consulting, Inc.

Coyote's Guide to IDL Programming: <http://www.dfanning.com/>

---